
内容梗概 
本研究では、Verilog HDLを用いて、マイクロプロセッサの最も基本的なアーキテク

チャを理解する目的で、マルチサイクルプロセッサの設計を行った。マイクロプロセッ

サの設計においては、慶応大学と東京工業大学で共同に開発した実験教育用の 16ビッ

ト幅の命令セットを持つ PICO16 があるが、これは SFL というHDL（ハードウェア

記述言語）を使用している。本研究ではこのアーキテクチャを理解した上で、Verilog 

HDLで実際に設計を行った。 

また、ハードウェア設計においてトップダウン設計が主流となっており、その設計方

法である仕様作成、動作レベルHDL 作成、動作レベルシミュレーション、RTLレベル

HDL 作成、論理合成、最適化、ゲートレベルシミュレーションまで行い、プロセッサ

の動作検証を行った。テストプログラムでは C 言語記述のシェルソートを PICO16で

の記述と比較することで、ハードウェアとソフトウェアの相互の理解を目的とする。 

1  はじめに 
 世界で最初のCPUは Intel社が1971年 11月に発表したマイクロプロセッサ「4004」

である。その動作周波数は 108KHz、トランジスタ数は 2300 個であった。これが 30

年後の現在では、Intel 社より「PentiumⅣ」が発表され、動作周波数は 3.06GHz、ト

ランジスタ数は 5500万個、20段のパイプラインなどが実装され、CPUの高速化が実

現されている。[8] 

1970年代に初めて IC（集積回路）が誕生してから、一般の人々に普及したのは電卓

や時計、ゲーム機といったパーソナル機器であった。1980年代に入ると、DRAMが普

及し、半導体の設計規模が増大した。また、従来は単体の機能だけを考慮して設計して

いたものが、家庭用電化製品や携帯電話に組み込まれているシステム全体を 1Chip 化

するといった SoC（System On Chip）設計などに変わってきており、設計規模が非常

に大きくなってきている。 

LSI設計フローは、従来はセル部から設計を始め、最終的に目的の製品を作成するボ

トムアップ設計という手法が用いられていたが、現在では機能の検証からアプローチす

るトップダウン設計に移行している。ボトムアップ設計では、設計が進むにつれ、修正

が困難になり、設計の最終段階まで製品の完全な動作が検証できないのに対し、トップ

ダウン設計では、ハードウェア記述言語を用い、論理シミュレーションによって設計の

初期で仕様の致命的な欠陥を発見でき、タイミングなどの考慮もする必要がなく、設計

時間の短縮化など、設計が容易にできるようになった。 

ハードウェア記述言語は VHDL や VerilogHDL などがあるが、本研究に用いる

VerilogHDLに関しては、1990年代にケイデンスが開発し、1995年 12月に IEEE1364 

に認定され、業界に浸透していくこととなった。また、C言語に似ていることや、動作

レベル、RTL レベル、ゲートレベルのすべての記述が同一の言語で実現できるなど、

非常に扱いやすい言語であると言える。また、FPGA（field programmable gate array）

の登場により、論理機能を安価に実際にチップに焼き付けることができるようになった

ため、開発コストの削減と、開発期間の短縮に繋がった。[1] 

今回設計した PICO16 は慶応義塾大学と東京工科大学で共同に開発されたマイクロ

プロセッサであり、16ビット RISC プロセッサである。命令長は 16ビットで、I形式、

R形式、J形式の３つの形式と、27個の命令を持つ。また PICO自体は SFL（Structured 

Functional  Language）というハードウェア記述言語を用い、NTT研究所が開発した

ハードウェア開発環境である PARTHENON（Parallel Architecture Refiner 

THEorized by Ntt Original Concept）を用いて設計されている。[3] 

本研究では、PICO16をVerolog HDLを用いて設計することにより、マイクロプロ

セッサの基本的なアーキテクチャを理解すし、HDL によるトップダウン設計の手順を

習得することを目的とする。ローム記念館の立命館大学VLSIセンターで用いられてい

る CADツールで設計した。VLSIセンターで実際に使用したツールは、NC-Verilog、

Verilog-XL（機能設計）、Design Compiler（論理合成）である。トップダウン設計に

基づいて、仕様作成、動作レベルHDL作成、動作レベルシミュレーション、RTLレベ

ル HDL作成、論理合成、最適化、ゲートレベルシミュレーションまで行った。 

第 2章ではハードウェア記述言語によるマイクロプロセッサの設計について述べる。

第 3 章では PICO16 マルチサイクルプロセッサのアーキテクチャについて述べる。第

４章では、ローム記念館の論理設計ツールである、NC-Verilog を用いた論理シミュレ

ーション、Design Compilerを用いた論理合成と最適化について述べる。第 5章ではC

言語記述のシェルソートを PICO 16で記述した上で、そのテストパターンのシミュレ

ーションによる検証、ハードウェア規模、性能などの検証と考察について述べる。 

２ ハードウェア記述言語によるマイクロプロセッサの設計 
２．１ ハードウェア記述言語 

 ハードウェア記述言語（HDL：Hardware Description Language）とはLSIやシス

テムなどの設計データを記述するための言語であり、ハードウェアの動作を論理ゲート

レベルより抽象度の高いレベルで記述できる言語である。それにより設計時間が短縮で

きることや、設計の変更や再利用が容易であり、設計者がより完成度の高いシステムを

設計することができる。またハードウェア記述言語はＣ言語に似ているため、言語の習

得と言う意味でも非常に有効である。 

代表的なHDLにはVHDL、VerilogHDL 、SFL（Structured Functional  Language）

などがあるが、ここでは VHDLと Verilog HDLの 2種類についてその特徴を示す。 

?? VHDL 

－米国防総省にて開発された言語 

－通信用 LSI、システム設計の開発に使用 

－文法的に厳格な言語 

?? VerilogHDL 

－パブリックドメインとしてケイデンスが公開 

－1995 年に IEEE1364として標準化 

－LSI設計の記述性を重視した言語 

また、図 1の全加算器における VerilogHDLでの記述を図２に示す。 

 

 

図 1：全加算器の論理回路図 

 

 

 

 

 

図 2：全加算器の VerilogHDLによる記述 

VerilogHDLでは 1行目でモジュール名の宣言と外部ポートの宣言を行う。1行目 2

行目の input、output で入出力信号の定義をする。そして、assign 文で実際の出力結

果と一致するようにプログラムを作成する。 

module fadder (a,b,cin,sum,cout); 
input a,b,cin; 
output sum,cout; 
 
assign cout = (a&b)|(b&cin)|(a&cin); 
assign sum  = (a&~b&~cin)|(~a&b&~cin)|(~a&~b&cin)|(a&b&cin);
 
endmodule 

２．２ トップダウン設計 

 ハードウェア記述言語の普及により従来のボトムアップ設計からトップダウン設計に移

行した。トップダウン設計とは図３のように抽象度の高い仕様設計から機能記述、回路図

を経て、実際の回路の設計に向かって徐々具体度を高めていく設計手法である。まず最初

に仕様を設計し、動作レベルでの HDLを記述し、テストパターンによって動作レベルシミ

ュレーションを行い、仕様通りの機能が得られるよう修正する。次に動作レベル HDL を

RTL（Register Transfar Level）HDLに変換し、それを論理合成、最適化した上で、ネッ

トリストを出力する。配置配線ツールなどによってゲートレベルシミュレーションを行い、

FPGA（Field Programmable Gate Array）などの Chipに焼き付けて動作検証を行う。 

 

 

図 3：トップダウン設計のフローチャート 

２．３ ＦＰＧＡ 

PLD（Programmable logic device）と呼ばれるプログラム可能な LSIは、SPLD（Simple 

PLD）、CPLD（Complex PLD）、FPGA（Field Programmable Gate Array）の 3 種類に

分類される。SPLD、CPLDは記録方式として EEEPROM構造をしており、またゲート数

は少ない。FPGA は SRAM 構造になっており、ゲート数は数百万ゲートで大規模である。

FPGA はプログラム書き換え可能なゲートアレイで、トップダウン設計により設計された

論理機能を焼き付けることで、その LSI が利用できる。従来のボトムアップ設計では大規

模な LSIの作成に多大なコストがかかっていたが、この FPGAによって動作検証ができる

ようになったことでこの問題は解決した。 

２．４ プロセッサアーキテクチャの分類 

 1971年 11月に Intel社が発表したマイクロプロセッサ「4004」が世界で最初のマイ

クロプロセッサである。現在では Intel 社より「PentiumⅣ」が発表されている。様々

なプロセッサが開発される中で命令実行方式にも違いがあり、そのアーキテクチャの分

類と、その動作について述べる。 

 
（１） 単一サイクル方式 

 単一サイクル方式では、一つの命令を１クロックサイクルで完了する。最も処理の長い

命令に依存するため、高速化ができない。現代のマイクロプロセッサに採用されていない。

また CPI は１となる。 

 

（２）マルチサイクル方式 

 マルチサイクル方式では、ひとつの命令を命令フェッチ、レジスタフェッチ、EX、メモ

リアクセス、などの複数のステップに分け、それぞれのステップが１クロックサイクルを

占める。また、単一サイクルに比べ CPI は増大するが、１クロックサイクルを短くするこ

とが可能で、クロック周波数をあげることができる。本研究で設計した PICO16 もマルチ

サイクル方式である。 

 

（３） パイプライン方式 

 パイプライン方式は命令の各実行ステップをステージに分割し、連続した各命令を少し

ずつずらして同時並列的に実行する方式である。命令のスループットが増大し、命令の全

体のクロックサイクルが大幅に減少する。だが、パイプラインステージを増やせば増やす

ほど高速化が期待できるわけではなく、実際は次のクロックサイクルで命令が実行できな

いというパイプラインハザードが起こる。パイプラインハザードにはデータハザード、制

御ハザードなどがあるが、これを考慮するため、制御も複雑になる。現在の Intel 社の

「PentiumⅣ」には２０段ものパイプラインが実装されスーパーパイプラインとも呼ば

れる。 

（４） スーパースカラ方式 

 スーパースカラ方式とは、フェッチやデコード、実行するためのユニットをそれぞれ複

数用意して、1 クロックサイクルで複数の命令を同時に実行する方式である。また命令実行

時にプロセッサ内部で動的に判断され、空いている実行ユニットがあれば自動的にそれを

使うようにスケジューリングが行われる。 

 

（５） VLIW方式 

 VLIW（Very Long Instruction Word）方式とは、ひとつの命令語中に、複数の命令を格

納しておき、それらをすべて同時に実行する方式である。常に決まった数の命令が同時に

パイプラインに投入され、同時に実行される。命令間には依存関係がないように事前にコ

ンパイラによって最適化する。VLIW の性能を最大限に引き出すためにはコンパイラによ

る最適化技術が欠かせない。 

３ マルチサイクルプロセッサ PICO16のアーキテクチャ 
３．１ PICO16の命令セット 

PICO16は、慶応義塾大学と東京工科大学で共同に開発した実験教育用のマイクロプ

ロセッサで、利用可能な FPGA のサイズに応じてさまざまなサイズの命令セットを構

築することができる。PICO16の命令セットは以下のような特徴をもつ。[3] 

?? Load/Store マシン（register-registerマシン）である。 
計算はレジスタ間でのみ許される。計算を行うためにはメモリからレジスタにデー

タをロードする必要がある。 

?? 16ビットの単一命令長を持つ 
命令長を固定することで、命令の作り方に苦しい部分がでてくるが、制御が圧倒的

に簡単になり、パイプライン化も容易になる。 

?? 単純な命令セット 
単純な命令セットを持つことで、簡単な実装ができる。 

 

PICO にはメモリアクセス命令を含むレジスタ間演算命令、immediate（即値）命令、

分岐命令の３種類の命令があり、図４に PICOの命令形式を示す。 

 

 5 3 3 5 

R形式 OP Rd Rs funct 

 5 3 8 

I形式 OP Rd immediate 

 5 11 

J形式 OP address 

図 4：PICOの命令形式 

 

各フィールドの意味を以下に示す 

OP（opcode）         ： 命令形式の判定 

Rd（destination Register）  ： 格納先レジスタ 

Rs（source Register）       ： 参照先レジスタ 

Function（functional code） ： 論理演算用の補助コード 

Ad d r e s s                     ： Rd に対する即値 

 

 PICO16 の命令セットはこれらの３つの命令形式をもち、I 形式（即値命令）は

immidiate 命令で、条件分岐命令などに使用する 8ビットのフィールドデータを格納す

る。R形式（レジスタ－レジスタ命令）は算術論理演算に利用する。J命令（分岐命令）

はジャンプ命令、サブルーチンコールなどに使用する。命令コード以外はすべてデータ

に使って、できる限り長いデータを格納する。また、PICO16の持つ命令数は 27個で

ある。その命令を表１に示す。 

表 1：PICO16の命令セット 

命令形式 命令コード  実行内容 

00000ddd_sss00000 NOP 有効な操作は行わない 

00000ddd_sss00001 MV sの値を dに格納 

00000ddd_sss00010 AND dと sの AND を dに格納 

00000ddd_sss00011 OR dと sの ORを dに格納 

00000ddd_sss00100 XOR dと sの XORを dに格納 

00000ddd_sss00101 NOT sの NOTを dに格納 

00000ddd_sss00110 ADD d＋sを dに格納 

00000ddd_sss00111 SUB d‐sをｄに格納 

00000ddd_sss01000 LD sで示す番地の値を dに格納 

00000ddd_sss01001 ST dで示す番地の値に sを格納 

00000ddd_sss01100 SL sの 1bit左シフトを dに格納 

R形式 

00000ddd_sss01101 SR sの１bit右シフトを dに格納 

00110ddd_xxxxxxxx ADDI d＋Xを dに格納 

00111ddd_xxxxxxxx SUBI d‐Xを dに格納 

00010ddd_xxxxxxxx ANDI dと Xの AND をｄに格納 

00011ddd_xxxxxxxx ORI dと Xの ORを dに格納 

00100ddd_xxxxxxxx XORI dと Xの XORを dに格納 

11100ddd_xxxxxxxx LDLI Xを dの下位 8bitに格納 

11101ddd_xxxxxxxx LDHI Xを dの上位 8bitに格納 

01000ddd_uuuuuuuu JALR 戻り番地を格納して dへ分岐 

01001ddd_xxxxxxxx BNEZ d≠0 ならば相対分岐 

01010ddd_xxxxxxxx BEQZ d=0 ならば相対分岐 

01011ddd_xxxxxxxx BMI d<0 ならば相対分岐 

I 形式 

01100ddd_xxxxxxxx BPL d? 0 ならば相対分岐 

01101_xxxxxxxxxxx JAL 戻り番地を格納して相対分岐 

01110ddd_uuuuuuuu JR dの内容に相対分岐 J 形式 

01111_xxxxxxxxxxx JMP 無条件相対分岐 

３．２ PICOのアーキテクチャ 

本研究で作成した PICO16のデータパスを図５に示す。 

 

 

図 5：PICO16のデータパス 

PICO16は表２に示す機能ユニットから構成されている。 

表 2：PICO16の機能ユニット 

プログラムカウンタ(pc) 次に読み込む命令を格納する 

命令レジスタ(ir) プログラムメモリから命令を格納する 

レジスタ(reg_file) 命令をデコードして格納する 

メモリアドレスレジスタ(mar) アクセスするメモリのアドレスを格納 

メモリデータレジスタ(mdr) メモリに書き込むデータを格納 

データメモリ(data memory) データを格納 

プログラムメモリ(program memory) 命令を格納 

ALU(alu)  算術演算を行う 

マルチプレクサ(mux) 入力するデータを選択する 

３．３ 命令の実行と制御 

PICO16はマルチサイクルプロセッサであり、命令フェッチと命令実行を繰り返しながら動

作する。メモリアクセス命令を含むレジスタ間演算命令、immediate（即値）命令、分

岐命令の３種類の命令があり、命令フェッチ、レジスタフェッチ、EX、EX2の 4つの

ステージに分かれ状態遷移を繰り返す。メモリアクセス命令と JAL命令のみが EX2を含

めた 4つのステージを実行し、その他の命令は EXまでの 3つのステージで実行する。
以下に４つの状態遷移を示す。 

 

?? IF：命令フェッチ 

プログラムアドレスである PCで指定されるプログラムメモリの値を irに格納する。 

 

?? RF：レジスタフェッチ 

Irのレジスタフィールド取ってきた命令中のオペランドフィールド(10～8bit、7～5bit)

で指定されるアドレスをレジスタから読み出し、セレクタ(mux1、mux2)に値を代入す

る。immidiate（即値）命令では、pc_addで下位 8 ビットの値を 16 ビット（上位 8 ビ

ットは 0）に拡張する。また 7～5bitは immidiateの数値の一部であるため、このアド

レスの値は利用されないが、読み出してもは問題はないため読み出しておく。また、こ

のステージで次の命令を読み出すための PCの値を ALUで１加算しておく。ST命令で

は読み出したレジスタの値をメモリデータレジスタ（mdr）に格納しておく。 

 

?? EX：実行１ 

メモリアクセス（LD、ST）以外の命令は EXステージで完了する。レジスタ間演算命

令では、セレクタから ALUに値を代入し、また irの 4～0bitは ALUのコマンド(com)

となっているため、そのコマンドを用いて演算を実行し dbusを経由してからレジスタ

ファイルに書き込む。Immidiate（即値）命令では、irの 15～11bitが ALUのコマン

ドとなっており、pc_addで符号拡張した値を演算する。LD命令ではアクセスするメモ

リのアドレスをメモリアドレスレジスタ（mar）に格納する。分岐命令では、判定する

条件の結果によって分岐先アドレスを pcに格納する。 

 

?? EX2：実行２（メモリアクセス） 

LD命令では、marに格納したアドレスを読み出し、レジスタに格納することで命令を

完了する。ST命令では、レジスタファイルから読み出したメモリのアドレスに mdrの

値を格納する。 

また、各命令の実行ステップを表３に示す。 

表 3：マルチサイクル方式ＰＩＣＯ16 の実行ステップ 

 R形式、I 形式 LD命令 ST命令 J 形式 JAL命令 

IF 命令フェッチ 

RF レジスタのフェッチ、プログラムカウンタの処理 

EX 
演算処理 メモリ 

アクセス 

メモリアドレス

を読み込む 

分岐処理 分岐元アドレス

を格納 

EX2 
 レジスタに

格納 

メモリに格納  分岐処理 

??演算処理：演算を実行して、その結果をレジスタに格納 

??分岐処理：分岐の条件による判断の結果を pcに格納 

４ VerilogHDLによるマルチサイクルプロセッサ PICO16の設計 
４．１ CAD ツールと設計手順 

本研究ではローム記念館のVLSIセンターで使用されているCADツールを用いて設計を

行った。使用したツールは以下の２つである。 

?? NC-verilog、Verilog-XL 

ケイデンス社の verilogシミュレータであり、NC-verilogはコンパイラ型の超高速イベ

ント・ドリブン・シミュレータであり、Verilog-XLはインタプリタ型の高速イベント・

ドリブン・シミュレータである。Verilog-XLは業界標準シミュレータとして確立されて

おり、多くの ASICベンダからサインオフ・シミュレータとして認定されている。また

NC-verilogと Verilog-XLは互換性があるため、本研究では NC-verilogを用いて、RTL

レベルシミュレーションとゲートレベルシミュレーションまでを行った。 

?? DesignCompiler 

Synopsys 社の論理合成、最適化ツールであり、verilog による回路記述と NC-verilog

による機能シミュレーションが完了したら，Design Compilerでターゲットの ASICラ

イブラリに基づいて論理合成を行い、最適化済みゲートレベル記述を作成する。 

 

本研究で行ったゲートレベルシミュレーションまでのトップダウン設計におけるツールの

領域を図 6 に示す。 

 

図 6：トップダウン設計におけるツールの領域 

また、このツールはローム記念館の端末でのみ使用可能である。VLSI センターの端末の

OSは solarisを用いている。NC-verilogによる各レベルシミュレーションの手順は端末エ

ミュレータから、 

% ncverilog +access+r test_pattern.v logic_ description.v –s +gui 

とコマンドを入力することで図７の GUI が立ち上がる。 

 

図 7：NC-Verilog 

また、これから波形を表示してシミュレーション結果を確認するには、図７の左側の

navigatorウィンドウで表示したい信号を選択してから右上の ボタンを押すことで、 
下図のような SignalscanWavesというグラフィック波形表示ツールが起動する。図８に表

示されている波形は図２の全加算器の verilogHDL記述より用いた。 

 

図 8：Signalscan Waves 

NC-verilogと Signal Wavesを用いて動作レベルの記述に問題がないことがわかれば、次に

Design Compilerを用いて論理合成と最適化を行う。Design Compilerの起動は、 

% design_analyser  

とコマンドを入力することで、Design CompilerのGUI版のdesign_analyserが起動する。

ここでは「File」→「Read」から全加算器の VerilogHDL 記述ファイルを読み込んだ。そ

のときに、ターゲットの ASICライブラリに基づいて回路の合成が行われる。 

最初にモジュールが２つ以上の場合は回路の平坦化、またはサブ回路の独自化を行う。回

路の平坦化は「Setup」→「Command Window」から開いたウィンドウに ungroup –all 

–flattenと入力して実行する。サブ回路の独自化は、トップモジュールを選択して「Edit」

→「Uniquify」→「Hierarchy」を実行する。 

次に入出力バッファセルの挿入を行う。まず入力信号の設定は、入力信号を選択して

「Attribute」→「Optimization Directives」→「Input port」から図９のような画面が開

く。Port is Padにチェックを入れ Port Pad Attributes...を押すとウィンドウが開く。そこ

では Exact Pad to Useにチェックをいれ、Pad Nameを VZIBUFとしてセルの挿入を実行

する。同様に出力信号の場合は、出力信号を選択して「Attribute」→「Optimization 

Directives」→「Output port」から図９と同様にして、Pad Name を VZOBUFとする。

次に、「Edit」→「Insert Pads」で開いたウィンドウをそのまま OKとしてこのセルの挿入

が完了する。そして「Tools」→「Design Optimization」で回路の最適化を行う。 

 

 

 

図 9：入力信号の設定 

次に図９の入出力ポートを全て選択し、「Attributes」→「Optimization Constraints」→

「Timing Constraints」を実行して、タイミングの制御の設定を行う。また、Clockがあれ

ば、「Attributes」→「Clocks」よりクロック周期の設定をする。今回の全加算器では Clock

がないのでこの設定はしない。これらの設定が完了した上で「Tools」→「Design 

Optimization」から回路の最適化をもう一度実行することで、design_analyzerによる回路

の最適化が完了する。図 10 にファイルを読み込んだときの全加算器の回路図（左）と、手

順を全て完了した後の回路図（右）を示す。 

 

図 10：回路の最適化 

 

 

 

 

４．２ モジュール構成 

図５のデータパスにより設計した PICO16 にはモジュールが、reg_fire、mux1、mux2、

ALU、mux3の５つがある。以下に各モジュールの入出力と動作を説明する。 

 

?? reg_file（レジスタファイル） 

命令を実行するためのレジスタのアドレスを受け取り各種演算命令の結果を出力する。

また、その結果を再び格納する。このプロセッサにおいて最も必要なモジュールのひと

つである。その入出力信号を表４に示す。 

表 4：reg_fireの入出力信号 

clk(１ビット) クロック信号、全ての動作が clkに同期する。 

rst(１ビット) clk の立ち上がりに１ならばレジスタファイル

の中身を初期化する。 

reg_req(１ビット) レジスタから読み込むときは１、書き込むとき

は０を入力する。 

reg_addr1(３ビット) 演算などに用いるレジスタファイルのアドレ

スである irの 10～8bitを入力する。 

reg_addr2(３ビット) レジスタファイルのアドレスである ir の 7～

5bitを入力する。 

入力信号 

reg_in(16ビット) 演算の結果や、分岐元アドレスの保存時にレジ

スタに入力する。 

reg_dout1(16ビット) 読み込んだ reg_addr1 のアドレスに入ってい

るレジスタのデータを出力する。 
出力信号 

reg_dout2(16ビット) 読み込んだ reg_addr2 のアドレスに入ってい

るレジスタのデータを出力する。 

 

?? mux1（データセレクタ１） 

mux1では、reg_fireから受け取った reg_dout1と pc（プログラムカウンタ）が入力さ

れ、どちらかを選択して出力する。ステージの初期値が与えられたときはクロックの立

ち下がりで pcを選択する信号が入力される。入出力信号を表５に示す。 

表 5：mux1の入出力信号 

mux1_sel(1ビット) １ならば mux1_in1 の値を出力し、０ならば

mux1_in2の値を出力する。 

mux1_in1(16ビット) レジスタファイルから出力された 

reg_dout1 のデータを入力する。 

入力信号 

mux1_in2(16ビット) pcの値を入力する。 

出力信号 mux1_out(16ビット) mux1_selの信号から値を出力する。 

 

 

?? mux2(データセレクタ２) 

mux2では reg_dout2とpcを加算する値または irから符号拡張された値が入力され、

どちらかを選択して出力する。mux1 と同様にステージの初期値が与えられたときは

クロックの立ち下がりで pcを加算する値を選択する信号が入力される。 



 17

表 6：mux2の入出力信号 

mux2_sel(1ビット) １ならば mux2_in1 の値を出力し、０ならば

mux2_in2の値を出力する。 

mux2_in1(16ビット) レジスタファイルから出力された 

reg_dout2 のデータを入力する。 
入力信号 

mux2_in2(16ビット) pcを加算する値や、immidiate命令で加算す

る数値を拡張して入力する。 

出力信号 mux2_out(16ビット) mux2_selの信号から値を出力する。 

 

特に mux2_outの出力が mux2_in2 の時は、表 7 のように命令によって入力される値

が変わる。 

表 7：各命令によるmux2_inの入力 

irに入力された命令 mux2_in2(16ビット)に入力される値 

LDLI、ADDI、SUBI ir[7]を上位 8 ビットに拡張、下位 8 ビットに ir[7:0] 

LDHI 上位 8 ビットに ir[7:0]、下位 8 ビットは 0 

ANDI、ORI、XORI 上位 8 ビットは 0、下位 8 ビットに ir[7:0] 

JAL、JMP 上位 5 ビットに ir[10]を拡張、下位 10 ビットに ir[10:0] 

BPL、BNEZ、BEQZ、BMI

が条件を満たす場合 
ir[7]を上位 8 ビットに拡張、下位 8 ビットに ir[7:0] 

上の命令が条件を満たさない

場合とステージが IFのとき 
16 進数「0001」 

 

 

?? ALU（論理演算ユニット） 

mux1と mux2のそれぞれからの出力を受け取り、irから受け取った opcodeによって

決められた演算を実行する。RFステージでは pcのカウントに使われ、EXステージで

は irから読み取った演算を実行する。 

表 8：ALUの入出力信号 

com(４ビット) opcodeが入力され、その codeによって行う演算

を決定する。 

reg1(16ビット) mux1の出力から入力される 
入力信号 

reg2(16ビット) mux2の出力から入力される 

出力信号 alu_out(16ビット) 演算結果を出力する。 

?? mux3（データセレクタ 3） 

alu からの出力を受け取り、また LD命令でロードする値を受け取る。 

表 9：mux3の入出力信号 

mux3_sel(1ビット) １ならば mux3_in1 の値を出力し、０ならば

mux3_in2の値を出力する。LD命令の時のみ

０となる。 

mux3_in1(16ビット) ALUから出力された alu_outが入力される。 
入力信号 

mux3_in2(16ビット) LD命令の時のみ、データメモリの値が入力さ

れる。 

出力信号 mux3_out(16ビット) mux3_selの信号から値を出力する。 

 

 

 
４．３ データ構成 

図５のデータパスにある中で、モジュールではないユニットの動作を以下に示す。 

 

?? pc（プログラムカウンタ） 

プログラムカウンタであり、次に読み込む命令のアドレスをプログラムメモリに送る。

また、pcの加算の時には mux1に出力される。 

?? program memory（プログラムメモリ） 

pc からプログラムアドレスを受け取り、prog_read の立ち上がりに ir へデータを出力

する。 

?? dbus 

mux3 から入力され、値を保持するためのバス、pc を加算した値が入力され、再び pc

に出力する。また ALUの演算結果も dbusに保持される。 

?? ir 

プログラムメモリから受け取ったデータを 10～8bitと、 7～5bit を reg_fileへ出力す

る。またプログラムカウンタで加算する値を mux2に出力する。 

?? mar（メモリアドレスレジスタ） 

LD命令時に reg_fileの reg_dout2 を受け取る。データメモリからロードするアドレス

が入力される。また ST命令時には dbusから格納するデータメモリのアドレスが入力

される。 

?? mdr（メモリデータレジスタ） 

ST命令時に reg_fileの reg_dout2 を受け取る。データメモリに格納するデータが入力

される。 

?? data memory 

dmem_reqが１の時は、ST命令時の mdrと marから受け取ったデータを書き込む。

dmem_reqが０の時は、LD命令時のmarから受け取ったアドレスを dmem_outとし

て、mux3に出力する。 

５ 設計した PICO16の検証 
５．１ シェルソートによるテストプログラムシミュレーション結果 

本研究で作成したテストプログラムは図 11 の C 言語記述によるシェルソートを用いた。

これを PICO16の命令セットで記述をして動作確認を行った。データ数は 5 個、20 個、

50 個の場合で行った。 

 

 

 

 

 

 

 

 

 

 

 

 

図 11：シェルソートの C言語記述 

またその C言語の 5 行を実際に PICO16の命令セットに変換した命令を図 12 に示す。 

C言語の for文の 1 行が PICO16の命令だと 18 行になっている。 

 

 

 

 

 

 

 

 

 

 

 

 

 

 

図 12：C言語の for文における PICO16命令記述 

1: void shell_sort(in a[], int n) 
2: { 
3:   int h, i, j, t; 
4:  
5:   for(h = 1; h < n/9; h * 3 + 1) 
6:   ; 
7:  
8:   for( ; h > 0; h = h/3){ 
9:       for(i = h; i < n; i++ ){ 
10:          j = i; 
11:          while (j >= h && a[j-h] > a[j]){ 
12:              t = a[j]; 
13:              a[j] = a[j-h];  
14:              a[j-h] = t; 
15:              j = j ? h; 
16:             } 
17:         } 
18:      } 
19:  } 
 

//5:   for(h = 1; h < n/9; h * 3 + 1) 
 
1:  11100000_00010100  //LDLI r1 #20   nの値 20を r1に格納 
2:  11100010_00000001  //LDLI r2 #1    hの初期値 1を r2 に格納
3:  11100101_00000000  //LDLI r5 #0    r5 に 0を格納 
4:  00111001_00001001  //SUBI r1 #9  r1 - 9 → r1 
5:  00110101_00000001  //ADDI r5 #1    r5 + 1 → r5 
6:  01100001_11111110  //BPL  r1 #-2   r1>0 ならば命令 4に分岐
7:  00111101_00000001  //SUBI r5 #1    r5 - 1 → r5 
8:  01010101_00001010  //BEQZ r5 #10   r5=0ならば命令 18に分岐
9:  11100100_00000010  //LDLI r4 #2    r4 に 2を格納 
10: 00000011_01000001  //MV   r3 r2    r2 を r3 にコピー 
11: 00000010_01100110  //ADD  r2 r3    r2 + r3 → r2 
12: 00111100_00000001  //SUBI r4 #1    r4 - 1 → r4 
13: 01001100_11111110  //BNEZ r4 #-2   r4≠0ならば命令 11に分岐
14: 00110010_00000001  //ADDI r2 #1    r2 + 1 → r2 
15: 00000110_10100001  //MV   r6 r5    r5 を r6 にコピー 
16: 00000110_01000111  //SUB  r6 r2    r6 - r2 → r6 
17: 01100110_11110111  //BPL  r6 #-8   r6>0 ならば命令 9に分岐
18: 00000111_01000001  //MV   r7 r2    r2 を r7 にコピー 

C言語と PICO16との命令数の比較を表 10 に示す。 

表 10：シェルソートの命令数の比較（データ 2０個） 

 C言語 PICO16  

静的命令数 19 55 

実行命令数 422 1436 

 

また、論理シミュレーションレベルでのクロックサイクルと実行命令数を表 11 に示す。 

ソート数が 5 個の時に CPI が下がっているが、これはデータの入れ替えが少なかったこ

とで、クロックサイクルが多い LD、ST 命令が少なかったからである。シェルソートす

るデータの並び方によって CPI 数が変化する。 

表 11：シェルソートの実行結果 

データ数 5 20 50 

実行命令数 226 1436 5164 

クロックサイクル数 716 4574 16460 

CPI  3.168 3.185 3.187 

 

５．２ プロセッサの評価 

４．１章で述べた design_analyzer を用いて最適化を行い、表 12 のような結果が得られ

た。入出力のポート数は変化しないが、特に配線の数が大幅に最適化されていることがわ

かる。 

表 12：PICO16の最適化 

 最適化前 最適化後 

ポート数 84 84 

配線数 1486 796 

セル（ゲート）数 763 671 

 

また、最適化が完了した後のゲート遅延時間、クロック周波数を表 13 に示す。 

表 13：最大ゲート遅延時間とクロック周波数 

最大ゲート遅延時間 クロック周波数 

38.98 ns 25.62 MHz 

５．３ シミュレーション結果の考察 

本研究では、マルチサイクルプロセッサである PICO16 を VerilogHDL で記述し、トッ

プダウン設計におけるゲートレベルシミュレーションまで行った。また最適化では、配線

数が大きく減少したことなどから、最適化の過程が理解できた。ゲートレベルシミュレー

ションは動作の確認は取れ、遅延などの情報も得ることができた。また、それを想定した

VerilogHDL記述ができるようになった。ただ、トップダウン設計の途中までしか完了し

ていないことを踏まえると、このマイクロプロセッサの設計はまだ完了したとは言えない

ので、VLSIセンターでのトップダウン設計を完全に理解する必要がある。 

シェルソートのテストプログラムでの実行結果から、C 言語による記述と PICO16 の命

令を比較して、C言語と命令語の違いを確認することができた。 

また PICO 16 では 27 命令あるが、2 つのレジスタのデータを比較して分岐する命令が

ないため、この命令を追加することでシェルソートなどの比較分岐が多いプログラムなど

には特に有効な効果が得られると考えられる。 

６ 終わりに 
本論文では、ハードウェア記述言語によるマルチサイクルプロセッサ PICO16 の設計につ

いて述べた。VLSIセンターのツールを使用して、論理レベルシミュレーションからゲート

レベルシミュレーションまでを行った。テストパターンはC言語のシェルソートを PICO16

の命令セットに記述したことにより、ハードウェアとソフトウェアの両方に対する理解を

深めることができた。 

現在ハードウェア記述言語が普及してきていることから、ハードウェア設計はよりソフト

ウェア開発に近いものになっている。その中で我々のようにソフトウェアを学んできた者

がハードウェアの知識を学ぶことで、ハードウェアとソフトウェアの双方のより深い知識

が得られ、双方のことを踏まえた開発ができる能力が身につくと信じている。 

今後の課題として、FPGA 上へ実際にダウンロードして動作検証を行い、トップダウン設

計の完全な習得を目指したいと考えている。 

付録 
１ マルチサイクルプロセッサ PICO16の VerilogHDL記述 
`timescale 1ns/100ps 
 
module 
new_pico(rst,clk,prog_in,pc,prog_read,data_addr,mdr,dmem_out,dmem_req); 
 
  input         rst, clk; 
  input  [15:0] prog_in,dmem_out; 
  output        dmem_req,prog_read; 
  output [15:0] pc,mdr,data_addr; 
   
  wire [15:0]  reg_dout1, reg_dout2; 
  wire [15:0]  dmem_out,reg_in,alu_out, 
        reg1,reg2,data_addr, 
        regi,prog_in; 
  wire [15:0]  mux2_in1,mux2_in2; 
  wire [3:0]   com; 
  wire [2:0]   addr1,reg_addr2; 
  wire         rst, clk; 
 
  reg  [15:0]  pc,dbus,mar,mdr,ir; 
  reg  [3:0]   stat; 
  reg         mux1_sel,mux2_sel,mux3_sel, 
        reg_req,dmem_req,prog_read; 
 
 
     
  parameter      ROP = 5'b00000, 
    LDLI= 5'b11100, 
    LDHI= 5'b11101, 
    BNEZ= 5'b01001, 
    BEQZ= 5'b01010, 
    BMI = 5'b01011, 
    BPL = 5'b01100, 
    JR  = 5'b01110, 
    JMP = 5'b01111, 
    JAL = 5'b01101, 
    ADDI= 5'b00110, 
    SUBI= 5'b00111, 
    ANDI= 5'b00010, 
    ORI = 5'b00011, 
    XORI= 5'b00100,    
    HALT= 5'b11111,     
    LD  = 5'b01000, 
    ST  = 5'b01001, 
    SL  = 5'b01100, 
    SR  = 5'b01101; 
   
  parameter      IF  = 4'b0001, 
    RF  = 4'b0010, 
    EX  = 4'b0100, 

    EX2 = 4'b1000; 
  alu16     alu1(.ina(reg1),  
                 .inb(reg2),  
   .com(com),  
   .y(alu_out)); 
   
  reg_file regf1(.reg_addr1(reg_addr1),  
   .reg_addr2(reg_addr2),  
   .reg_in(reg_in),  
   .reg_dout1(reg_dout1),  
   .reg_dout2(reg_dout2),  
   .reg_req(reg_req), 
   .rst(rst),   
   .clk(clk)); 
 
  mux1       mu1(.mux1_in1(reg_dout1), 
   .mux1_in2(pc), 
   .mux1_out(reg1),  
   .mux1_sel(mux1_sel)); 
    
 
  mux2       mu2(.mux2_in1(mux2_in1), 
   .mux2_in2(mux2_in2), 
   .mux2_out(reg2),  
   .mux2_sel(mux2_sel)); 
  
  mux3       mu3(.mux3_in1(alu_out), 
   .mux3_in2(dmem_out), 
   .mux3_out(regi),  
   .mux3_sel(mux3_sel)); 
    
  assign    com = ((ir[15:11] == ROP &&  
    ir[4:0] != ST &&  
    ir[4:0] != LD) &&  
    rst == 0 &&  
    stat != IF)? ir[3:0]: 
                 ((ir[15:11] == ADDI ||  
           ir[15:11] == SUBI || 
           ir[15:11] == ANDI || 
                 ir[15:11] == ORI  || 
                  ir[15:11] == XORI) &&  
    rst == 0 &&  
    stat != IF)? ir[14:11]:  
          (rst == 1)? 4'bx: 
          (stat == IF)? 4'b0110: 
          ((ir[15:11] == BPL  ||     
    ir[15:11] == BNEZ || 
    ir[15:11] == BEQZ || 
    ir[15:11] == BMI ||  
           ir[15:11] == JMP) &&  
    stat != IF)? 4'b0110:  
          ((ir[15:11] == LDLI || 
    ir[15:11] == LDHI || 
    ir[15:11] == JAL) && 

    stat != IF)? 4'b0001: 4'b0000; 
   
  assign    mux2_in1 = (rst == 0)? reg_dout2: 16'bx; 
   
  assign    mux2_in2 = ((ir[15:11] == LDLI || 
          ir[15:11] == ADDI ||  
                 ir[15:11] == SUBI) && 
         rst == 0 &&  
         stat != IF)?  
    {ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7:0]}: 
                (ir[15:11] == LDHI && 
          rst == 0 &&  
          stat != IF)? {ir[7:0],8'h00}:  
               ((ir[15:11] == ANDI || 
                       ir[15:11] == ORI  || 
                        ir[15:11] == XORI) && 
         rst == 0 &&  
         stat != IF)? {8'h00,ir[7:0]}: 
               (((ir[15:11] == BPL &&  
          reg_dout1[15] == 1'h0)|| 
         (ir[15:11] == BNEZ &&  
         reg_dout1 != 16'h0000) || 
         (ir[15:11] == BEQZ &&  
                reg_dout1 == 16'h0000) || 
         (ir[15:11] == BMI &&  
                   reg_dout1[15] == 1'b1)) &&  
         rst == 0 &&  
         stat != IF)?  
    {ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7],ir[7:0]}: 
                       ((ir[15:11] == JAL || 
          ir[15:11] == JMP) && 
                rst == 0 &&  
         stat != IF)?  
    {ir[10],ir[10],ir[10],ir[10],ir[10],ir[10:0]}:  
              (((ir[15:11] == BPL &&  
          reg_dout1[15] == 1'b1) || 
         (ir[15:11] == BNEZ &&  
          reg_dout1 == 16'h0000) || 
         (ir[15:11] == BEQZ &&  
                 reg_dout1 != 16'h0000) || 
         (ir[15:11] == BMI &&  
                    reg_dout1[15] != 1'b1)) || 
          stat == IF)? 16'h0001:  16'bx; 
   
  
  assign    reg_in = (rst == 0)? dbus: 16'bx; 
   
  assign    data_addr = (ir[15:11] == ROP &&  
          ir[4:0] == ST && 
          rst == 0 && stat != IF)? regi: 
                              (ir[15:11] == ROP &&  
          ir[4:0] == LD && 
          rst == 0 && stat != IF)? mar:   
                 16'bx; 

  assign    reg_addr1 = (rst == 0)? ir[10:8]: 3'bx;  
  assign    reg_addr2 = (rst == 0)? ir[7:5]: 3'bx; 
 
  
  always @(posedge clk) begin                /* STATE */ 
 
    if(rst == 1) begin 
      stat <= 4'b0000; 
    end 
     
    if(stat == 4'b0000 && rst == 0) begin        
      stat <= IF; 
      ir <= prog_in; 
    end 
     
    if(stat == IF && ir[15:11] == HALT && rst == 0) begin 
      stat <= 4'b1111; 
      end 
     
    if(stat == IF && ir[15:11] != HALT && rst == 0) begin   
      stat <= RF; 
    end 
     
    if(stat == RF && rst == 0) begin    
      stat <= EX; 
    end   
     
    if(stat == EX && rst == 0) begin 
      if((ir[15:11] == ROP && ir[4:0] == LD) || 
  (ir[15:11] == ROP && ir[4:0] == ST) || 
  ir[15:11] == JAL) begin 
 stat <= EX2; 
      end   
      else begin 
 stat <= IF; 
 ir <= prog_in; 
      end 
    end 
   
    if(stat == EX2 && rst == 0) begin 
      stat <= IF; 
      ir <= prog_in; 
    end   
  end 
 
   
  always @(negedge clk)                    /* REQUIRE */ 
    begin 
       
      if(rst == 1) begin 
 pc  <= 16'h0000; 
 prog_read <= 1'b0;   
 dmem_req <= 1'b0; 

      end 
       
      if(stat == 4'b0000 &&  
  rst == 0) begin 
        mux2_sel <= 1'b0; 
 mux1_sel <= 1'b0; 
 reg_req <= 1'b0; 
        prog_read <= 1'b1; 
 mux3_sel <= 1'b1; 
      end 
            
      if(stat == IF && rst == 0 &&  
  (ir[15] == 0  || ir[15] == 1))          /*IF stage*/ 
 begin 
   prog_read <= 1'b0; 
    if(ir[15:11] == ADDI ||  
    ir[15:11] == SUBI || 
    ir[15:11] == ANDI || 
          ir[15:11] == ORI  || 
           ir[15:11] == XORI ) begin 
     dbus <= regi; 
     mux2_sel <= 1'b0; 
            mux1_sel <= 1'b1; 
           end 
   else if(ir[15:11] == LDLI || 
    ir[15:11] == LDHI || 
    ir[15:11] == JAL) begin 
     dbus <= regi; 
     end 
   else if(ir[15:11] == ROP &&  
    ir[4:0] == LD)  begin 
     dbus <= regi; 
     mux2_sel <= 1'b1; 
            mux1_sel <= 1'b1; 
   end 
 
   else if (ir[15:11] != BPL  &&  
     ir[15:11] != BNEZ && 
     ir[15:11] != BEQZ && 
     ir[15:11] != BMI  &&   
     ir[15:11] != JMP  && 
     ir[15:11] != JR)  begin 
     dbus <= regi; 
     mux2_sel <= 1'b1; 
     mux1_sel <= 1'b1; 
     reg_req <= 1'b0; 
   end 
    
 end 

      if(stat == RF && rst == 0) begin          /*RF stage*/ 
  
 if (ir[15:11] == BPL  ||     
     ir[15:11] == BNEZ || 
     ir[15:11] == BEQZ || 
     ir[15:11] == BMI  || 
     ir[15:11] == JMP  || 
     ir[15:11] == JR) begin 
   dbus <= regi; 
        end     
 else if(ir[15:11] == JAL) begin 
         reg_req <= 1'b1;    
 end  
 else if(ir[15:11] == ADDI ||  
  ir[15:11] == SUBI || 
  ir[15:11] == ANDI || 
  ir[15:11] == ORI  || 
  ir[15:11] == XORI) begin  
   reg_req <= 1'b1; 
   pc <= dbus; 
   dbus <= regi; 
 end 
        else if(ir[15:11] == ROP &&  
  ir[4:0] == LD) begin 
   mux3_sel <= 1'b0; 
         dmem_req <= 1'b0; 
   mar <= reg_dout2; 
   pc <= dbus; 
 end  
 else if(ir[15:11] == ROP &&  
  ir[4:0] == ST) begin 
   mdr <= reg_dout2; 
   pc <= dbus; 
 end 
 else begin 
   pc <= dbus; 
   dbus <= regi; 
   reg_req <= 1'b1; 
 end  
      end // if (stat == RF) 
       
       
       
      if(stat == EX && rst == 0)  begin        /*EX stage*/ 
 if(ir[15:11] == ROP) begin   
   if(ir[4:0] == LD ) begin 
    // pc <= dbus; 
     dbus <= regi; 
     reg_req <= 1'b1; 
   end 
   else if(ir[4:0] == ST) begin 
     pc <= dbus; 
     dmem_req <= 1'b1; 

   end 
   else begin 
            mux1_sel <= 1'b0; 
            mux2_sel <= 1'b0; 
     reg_req  <= 1'b0; 
     prog_read <= 1'b1;        
   end 
 end 
 else if(ir[15:11] == ADDI ||  
  ir[15:11] == SUBI || 
  ir[15:11] == ANDI || 
  ir[15:11] == ORI  || 
  ir[15:11] == XORI)  
   begin 
            mux2_sel <= 1'b0; 
     mux1_sel <= 1'b0; 
     reg_req <= 1'b0; 
     prog_read <= 1'b1;  
   end       
 else if(ir[15:11] == BNEZ || 
  ir[15:11] == BEQZ || 
  ir[15:11] == JMP || 
  ir[15:11] == JR  || 
  ir[15:11] == BPL || 
  ir[15:11] == BMI)  
   begin 
     pc <= dbus; 
     mux1_sel <= 1'b0; 
            mux2_sel <= 1'b0; 
     prog_read <= 1'b1;      
   end 
 else if(ir[15:11] == JAL) begin 
   reg_req <= 1'b0; 
   dbus <= regi; 
 end 
 else if(ir[15:11] == LDLI || 
  ir[15:11] == LDHI) begin 
   reg_req <= 1'b0; 
   prog_read <= 1'b1; 
          mux2_sel <= 1'b0; 
   mux1_sel <= 1'b0; 
 end 
      end // if (stat == EX && rst == 0) 
       
 
       
      if(stat == EX2 && rst == 0)begin          /*EX2 stage*/ 
 prog_read <= 1'b1; 
        if(ir[15:11] == ROP &&  
    ir[4:0] == LD) begin 
   mux2_sel <= 1'b0; 
          mux1_sel <= 1'b0; 
   mux3_sel <= 1'b1; 
   reg_req <= 1'b0; 

 end  
 else if(ir[15:11] == ROP &&  
  ir[4:0] == ST) begin 
          dmem_req <= 1'b0; 
          mux1_sel <= 1'b0; 
   mux2_sel <= 1'b0; 
 end 
       else if(ir[15:11] == JAL) begin 
          mux1_sel <= 1'b0; 
   mux2_sel <= 1'b0; 
   reg_req  <= 1'b0; 
   pc <= dbus; 
 end 
      end     
    end // always @ (negedge clk) 
   
endmodule 
 
 
module alu16(ina, inb, com, y); 
 
  input  [15:0] ina, inb; 
  input  [3:0]  com; 
  output [15:0] y; 
 
  wire [15:0]  ina, inb, y; 
  wire [3:0]    com; 
 
  assign  y = (com == 4'b0000)? ina: 
                    (com == 4'b0001)? inb: 
                    (com == 4'b0010)? ina & inb: 
                    (com == 4'b0011)? ina | inb: 
                    (com == 4'b0100)? ina ^ inb: 
                    (com == 4'b0101)? ~inb: 
                    (com == 4'b1100)? inb << 1: 
                    (com == 4'b1101)? inb >> 1: 
                    (com == 4'b0110)? ina + inb: 
                    (com == 4'b0111)? ina - inb:16'bx;          
 
endmodule 
 
 
module mux3(mux3_in1,mux3_in2,mux3_out,mux3_sel); 
 
  input  [15:0] mux3_in1,mux3_in2; 
  input         mux3_sel; 
  output [15:0] mux3_out; 
   
  wire [15:0]  mux3_out,mux3_in1,mux3_in2; 
  wire   mux3_sel; 
 
  assign  mux3_out = (mux3_sel == 1'b1)? mux3_in1: 
                           (mux3_sel == 1'b0)? mux3_in2: 16'bx; 

endmodule // mux3 
 
 
 
module mux2(mux2_in1,mux2_in2,mux2_out,mux2_sel); 
 
  input  [15:0] mux2_in1,mux2_in2; 
  input         mux2_sel; 
  output [15:0] mux2_out; 
   
  wire [15:0]  mux2_out,mux2_in1,mux2_in2; 
  wire       mux2_sel; 
 
  assign      mux2_out = (mux2_sel == 1'b1)? mux2_in1: 
                               (mux2_sel == 1'b0)? mux2_in2: 16'bx; 
 
endmodule // mux2 
 
module mux1(mux1_in1,mux1_in2,mux1_out,mux1_sel); 
 
  input  [15:0] mux1_in1,mux1_in2; 
  input         mux1_sel; 
  output [15:0] mux1_out; 
   
  wire [15:0]  mux1_out,mux1_in1,mux1_in2; 
  wire   mux1_sel; 
   
  assign  mux1_out = (mux1_sel == 1'b1)? mux1_in1: 
                           (mux1_sel == 1'b0)? mux1_in2: 16'bx; 
 
endmodule // mux1 
 
 
module reg_file (reg_addr1, reg_addr2, reg_in, reg_dout1, reg_dout2, 
reg_req,rst,clk); 
 
  input  [2:0]  reg_addr1, reg_addr2; 
  input  [15:0] reg_in; 
  input         reg_req,rst,clk; 
  output [15:0] reg_dout1, reg_dout2; 
 
  reg  [15:0]  register0, register1,  
  register2, register3,  
  register4, register5,  
  register6, register7; 
  wire   reg_req,rst,clk; 
  wire [2:0]  reg_addr1, reg_addr2; 
  wire [15:0]  reg_dout1, reg_dout2,reg_in; 
 
   
// read req=0 
  assign    reg_dout1 = (reg_addr1 == 3'b000 &&  
    reg_req == 0)? register0: 
                        (reg_addr1 == 3'b001 &&  

    reg_req == 0)? register1: 
                        (reg_addr1 == 3'b010 && 
    reg_req == 0)? register2: 
                        (reg_addr1 == 3'b011 &&  
    reg_req == 0)? register3: 
                        (reg_addr1 == 3'b100 &&  
    reg_req == 0)? register4: 
                        (reg_addr1 == 3'b101 &&  
    reg_req == 0)? register5: 
                        (reg_addr1 == 3'b110 &&  
    reg_req == 0)? register6: 
                        (reg_addr1 == 3'b111 && 
    reg_req == 0)? register7:16'bx; 
 
  assign    reg_dout2 = (reg_addr2 == 3'b000 &&  
    reg_req == 0)? register0: 
                        (reg_addr2 == 3'b001 &&  
    reg_req == 0)? register1: 
                        (reg_addr2 == 3'b010 && 
    reg_req == 0)? register2: 
                        (reg_addr2 == 3'b011 &&  
    reg_req == 0)? register3: 
                        (reg_addr2 == 3'b100 &&  
    reg_req == 0)? register4: 
                        (reg_addr2 == 3'b101 &&  
    reg_req == 0)? register5: 
                        (reg_addr2 == 3'b110 &&  
    reg_req == 0)? register6: 
                        (reg_addr2 == 3'b111 && 
    reg_req == 0)? register7:16'bx; 
 
//write reg_req=1 
  always @(posedge clk) 
    begin 
      if(reg_req) begin 
 case (reg_addr1) 
   3'b000: register0 = reg_in; 
   3'b001: register1 = reg_in; 
   3'b010: register2 = reg_in; 
   3'b011: register3 = reg_in; 
   3'b100: register4 = reg_in; 
   3'b101: register5 = reg_in; 
   3'b110: register6 = reg_in; 
   3'b111: register7 = reg_in; 
 endcase // case(reg_addr1) 
      end 
 
      if(rst == 1) begin 
 register0 = 16'b00000000_00000000; 
 register1 = 16'b00000000_00000000; 
 register2 = 16'b00000000_00000000; 
 register3 = 16'b00000000_00000000; 
 register4 = 16'b00000000_00000000; 
 register5 = 16'b00000000_00000000; 

 register6 = 16'b00000000_00000000; 
 register7 = 16'b00000000_00000000; 
      end       
    end 
 
  
endmodule // reg_file 
 
２ シェルソートテストプログラム 

 
0:11100000_00010100   //LDLI r0 #20 

 1:00000001_00000001   //MV   r1 r0 
 2:11100010_00000001   //LDLI r2 #1 
 3:11100101_00000000   //LDLI r5 #0 
 4:00111001_00001001   //SUBI r1,#9 
 5:00110101_00000001   //ADDI r5 #1 
 6:01100001_11111110   //BPL  r1,#-2 
 7:00111101_00000001   //SUBI r5,#1 
 8:01010101_00001010   //BEQZ r5 #10 
 9:11100100_00000010   //LDLI r4 #2 
10:00000011_01000001   //MV   r3 r2 
11:00000010_01100110   //ADD  r2,r3 
12:00111100_00000001   //SUBI r4,#1 
13:01001100_11111110   //BNEZ r4 #2 
14:00110010_00000001   //ADDI r2 #1 
15:00000110_10100001   //MV   r6 r5 
16:00000110_01000111   //SUB  r6,r2 
17:01100110_11110111   //BPL  r6,#-8 
18:00000111_01000001   //MV   r7 r2  
19:00000010_11100001   //MV   r2 r7 
20:01010010_00100011   //BEQZ r2 #35 
21:01111000_00001001   //JMP  #+9 
22:11100110_00000000   //LDLI r6 #0 
23:11100001_00000011   //LDLI r1 #3 
24:00000010_00100111   //SUB  r2,r1 
25:00110110_00000001   //ADDI r6 #1 
26:01100010_11111110   //BPI  r2,#-2 
27:00111110_00000001   //SUBI r6,#1 
28:00000010_11000001   //MV   r2 r6 
29:01100010_11110111   //BMI  r2,#-8 
30:00000011_01000001   //MV   r3 r2 
31:00000001_00000001   //MV   r1 r0 
32:00000001_01100111   //SUB  r1 r3 
33:01011001_11110101   //BMI  r1 #-11 
34:01010001_11110100   //BEQZ r1 #-12 
35:00000100_01100001   //MV   r4 r3 
36:00000101_10000001   //MV   r5 r4 
37:00000100_01000111   //SUB  r4 r2 
38:01011100_00001111   //BMI  r4 #15 
39:00000110_10101000   //LD   r6 (r5) 
40:00000111_10001000   //LD   r7 (r4) 

41:00000001_11100001   //MV   r1 r7 
42:00000001_11000111   //SUB  r1 r6 
43:01011001_00001010   //BMI  r1 #10 
44:01010001_00001001   //BEQZ r1 #9 
45:00000001_11000001   //MV   r1 r6 
46:00000110_11100001   //MV   r6 r7 
47:00000111_00100001   //MV   r7 r1 
48:00000101_11001001   //ST  (r5) r6 
49:00000100_11101001   //ST  (r4) r7 
50:00000100_10100001   //MV   r4 r5 
51:00000100_01000111   //SUB  r4 r2 
52:01111111_11110000   //JMP  #-16  
53:00110011_00000001   //ADDI r3 #1 
54:01111111_11101001   //JMP  #-23 
55:11111111_11111111    //HALT 


